
ΑΣΚΗΣΗ 3: ΠΟΛΛΑΠΛΑ ΣΥΜΒΟΛΑ & ΔΟΜΕΣ ΓΕΓΟΝΟΤΩΝ
ΠΕΡΙΣΣΟΤΕΡΑ ΓΙΑ ΤΗΝ ΧΡΗΣΗ ΤΟΥ ΧΑΡ/ΡΑ ΜΠΑΛΑΝΤΕΡ (?)

Στο προηγούμενο εργαστήριο είδαμε την αντικατάσταση του συμβόλου ? από οποιοδήποτε σύμβολο στην υπόθεση ενός κανόνα. Υποθέστε ότι έχουμε τα εξής γεγονότα:
(member-of beatles john_lennon paul_mccartney george_harrison ringo_starr)

(member-of who roger_daltrey pete_townsend keith_moon)

(member-of ebtg tracey_thorn ben_watt)
Εάν θέλουμε να γράψουμε έναν κανόνα που θα ενεργοποιείται και από τα τρία αυτά γεγονότα δεν μπορούμε να χρησιμοποιήσουμε τον χαρακτήρα ?, γιατί θα αντικαταστήσει μόνο ένα σύμβολο. Θα χρησιμοποιήσουμε τον συνδυασμό χαρακτήρων $?, που μπορεί να αντικαταστήσει κανένα ή περισσότερα σύμβολα:

(defrule bands

  (member-of ?band $?)

=>

  (printout t "there is a band called " ?band crlf))
Ο παραπάνω κανόνας θα έχει ως αποτέλεσμα: 
CLIPS>(run)

there is a band called ebtg

there is a band called beatles

there is a band called who

Σε ένα επόμενο βήμα μπορούμε να πάρουμε μια λίστα με τα μέλη των συγκροτημάτων:
(defrule band-members

  (member-of ?band $? ?member $?)

=>

  (printout t ?member " is a member of " ?band crlf))
Στην υπόθεση αυτού του κανόνα, οι χαρακτήρες $? θα αντικαταστήσουν κανένα ή περισσότερα σύμβολα. Η μεταβλητή ?member θα πάρει μια τιμή κάθε φορά (θα αντικατασταθεί με ένα σύμβολο κάθε φορά). Έτσι, η ?member θα ταιριάξει μια φορά με κάθε μέλος συγκροτήματος, ενώ οι συνδυασμοί $? θα ταιριάξουν με τα υπόλοιπα μέλη που προηγούνται (ο πρώτος συνδυασμός) ή έπονται (ο δεύτερος συνδυασμός). Για παράδειγμα, ο επόμενος πίνακας δείχνει όλους τους τρόπους με τους οποίους ο κανόνας θα ταιριάξει με το γεγονός των Beatles:

	
	Ταίριασμα πρώτου $?
	Ταίριασμα $member 
	Ταίριασμα τελευταίου $?

	1
	τίποτα
	john_lennon
	paul_mccartney george_harrison ringo_starr

	2
	john_lennon
	paul_mccartney 
	george_harrison ringo_starr

	3
	john_lennon  paul_mccartney 
	george_harrison 
	ringo_starr

	4
	john_lennon  paul_mccartney george_harrison 
	ringo_starr
	τίποτα


Είναι επίσης δυνατή η χρήση μεταβλητών πολλαπλών τιμών: 
 (defrule band-members

  (member-of ?band $?members)

=>

  (printout t "The members of " ?band " are " $?members crlf))
ΠΕΡΙΣΣΟΤΕΡΑ ΓΙΑ ΤΙΣ ΜΕΤΑΒΛΗΤΕΣ
Μέχρι τώρα είδαμε την ενσωμάτωση μεταβλητών μόνο στην υπόθεση ενός κανόνα. Με την συνάρτηση (bind), είναι δυνατή η δημιουργία μιας προσωρινής μεταβλητής στο δεξιό τμήμα ενός κανόνα:

(defrule addup

  (number ?x)

  (number ?y)

=>

  (bind ?total (+ ?x ?y))

  (printout t ?x " + " ?y " = " ?total crlf)

  (assert (total ?total)))
Έχετε υπόψη σας ότι η προσωρινή μεταβλητή ?total έχει διάρκεια ζωής μόνο μέσα στον κανόνα. Δεν μπορείτε να την χρησιμοποιήσετε έξω από τον κανόνα. Εάν χρειάζεστε μεταβλητές που μπορούν να χρησιμοποιηθούν από περισσότερους κανόνες χωρίς να χάνουν την τιμή τους, πρέπει να τις δηλώσετε χρησιμοποιώντας την δομή (defglobal). Για παράδειγμα:
(defglobal

  ?*var1* = 17

  ?*oranges* = "seven"

)
Μετά από ένα (reset), θα δημιουργηθούν δύο καθολικές μεταβλητές, οι ?*var1* και ?*oranges* (οι αστερίσκοι είναι απαραίτητοι), με τιμές 17  και "seven" αντίστοιχα.  Αυτές οι δύο μεταβλητές μπορούν να χρησιμοποιηθούν από οποιονδήποτε κανόνα, και η τιμή τους μπορεί να αλλάξει με την συνάρτηση (bind).
Άλλο παράδειγμα
CLIPS> (defglobal ?*x* = 3.4)

CLIPS> ?*x*

3.4

CLIPS> (bind ?*x* (+ 8 9))

17

CLIPS> ?*x*

17

CLIPS> (bind ?*x* (create$ a b c d))

(a b c d)

CLIPS> ?*x*

(a b c d)

CLIPS> (bind ?*x* d e f)

(d e f)

CLIPS> ?*x*

(d e f)

CLIPS> (bind ?*x*)

3.4

CLIPS> ?*x*

3.4

CLIPS>

ΔΟΜΕΣ ΓΕΓΟΝΟΤΩΝ (TEMPLATES)
Με απλά γεγονότα μπορεί κανείς να πετύχει πολλά, όμως πολλές φορές χρειάζεται να συνδυαστεί πληροφορία περισσότερων γεγονότων. Φανταστείτε μια εφαρμογή που προσπαθεί να βρει ανθρώπους που ταιριάζουν μεταξύ τους, σύμφωνα με διάφορα χαρακτηριστικά, τα οποία θα παίρνουν τιμή για τον κάθε άνθρωπο. Έτσι για να περιγράψουμε δύο ανθρώπους θα γράφαμε:
(age Andrew 20)

(weight Andrew 80)

(height Andrew 188)

(blood-pressure Andrew 130 80)

(age brenda 23)

(weight brenda 50)

(height brenda 140)

(blood-pressure brenda 120 60)

Αυτά είναι πολλά και διαφορετικά γεγονότα, και τίποτα δεν συνδέει τα χαρακτηριστικά του ίδιου ανθρώπου μαζί εκτός από ένα και μόνο κοινό πεδίο με το όνομα του ανθρώπου. Ένας καλύτερος τρόπος να αναπαραστήσουμε αυτή την πληροφορία είναι με μια δομή γεγονότων, με την εντολή (deftemplate). Έτσι:

(deftemplate personal-data


(slot name)


(slot age)


(slot weight)


(slot height)


(multislot blood-pressure)

)

Η (deftemplate) δεν δημιουργεί τα γεγονότα. Δημιουργεί την μορφή που θα πάρουν τα γεγονότα. Κάθε φορά που δημιουργείται ένα γεγονός αυτής της δομής, περιέχει τα πεδία (slots) που ορίζονται με τον παραπάνω ορισμό της δομής του γεγονότος. Κάθε πεδίο μπορεί να περιέχει μια τιμή και μπορεί να προσπελαστεί με το όνομά του. Με την εντολή (assert) μπορούν τώρα να δημιουργηθούν τα γεγονότα με την πληροφορία κάθε ανθρώπου:

(assert (personal-data (name Andrew) (age 20) (weight 80) 


(height 188) (blood-pressure 130 80)))

Μπορεί να χρησιμοποιηθεί και η (deffacts):

(deffacts people


(personal-data (name Andrew) (age 20) (weight 80) 



(height 188) (blood-pressure 130 80))


(personal-data (name Cyril) (age 63) (weight 70)



(height 1678) (blood-pressure 180 90)))

Δεν είναι απαραίτητο να δώσω πληροφορία για όλα τα πεδία, ούτε έχει σημασία η σειρά με την οποία τοποθετούνται τα πεδία. Έτσι η εντολή 
(assert (personal-data (weight 150) (age 23) (name Brenda)))

είναι έγκυρη. 
Οι δομές γεγονότων μπορούν να αλλάξουν χωρίς να χρειάζεται να αφαιρεθούν και να ξαναδημιουργηθούν στην καινούρια έκδοση. Η συνάρτηση (modify) επιτρέπει την μεταβολή ενός ή περισσοτέρων πεδίων σε ένα γεγονός.  Υποθέστε ότι είναι τα γενέθλια του Andrew. Αν ορίσουμε τον κανόνα 

(defrule birthday


?birthday <- (birthday ?name)


?data-fact <- (personal-data (name ?name) (age ?age))

=>


(modify ?data-fact (age (+ ?age 1)))


(retract ?birthday)

)

και εισάγουμε το γεγονός (birthday Andrew), τότε η ηλικία του Andrew θα αυξηθεί κατά 1, ενώ όλα τα υπόλοιπα στοιχεία του θα παραμείνουν αμετάβλητα. Το γεγονός (birthday Andrew) στην συνέχεια αφαιρείται για να εξασφαλιστεί ότι δεν θα πυροδοτηθεί ο κανόνας σύντομα ξανά, και αυξηθεί η ηλικία του Andrew με ταχύτατους ρυθμούς. 
Οι δομές δεδομένων μπορούν να χρησιμοποιηθούν σε κανόνες, όπως όλα τα γεγονότα:
(defrule lardy-bugger


(personal-data (name ?name) (weight ?weight))


(test (> ?weight 100))

=>


(printout t ?name " weighs " ?weight " kg - the fat sod." crlf)

)

Η εντολή (test (> ?weight 100)) είναι ένα στοιχείο υπόθεσης (conditional element). Επιτρέπει στην υπόθεση (το αριστερό μέρος) του κανόνα να κάνει έλεγχο μιας έκφρασης. Στην συγκεκριμένη περίπτωση θα ταιριάξει με γεγονότα των οποίων η τιμή του πεδίου weight ξεπερνάει τα 100 kg. 
Θυμηθείτε ότι περισσότερα από ένα γεγονότα μπορούν να συνδεθούν στην υπόθεση ενός κανόνα είτε με τον τελεστή AND
(defrule print-ages


(and 



(personal-data (name ?name) (age ?age))



(personal-data (name ?name) (weight ?weight))


)

=>


(printout t ?name " weighs " ?weight " at " ?age " years old." crlf)

)
είτε με τον τελεστή OR

(defrule take-an-umbrella


(or 



(weather raining)



(weather snowing)


)

=>


(printout t "Take an umbrella" crlf)

)

To AND είναι περιττό (εννοείται) και μπορεί να παραλειφθεί. 
Με το στοιχείο test μπορώ να ελέγξω οτιδήποτε στο αριστερό τμήμα ενός κανόνα. Έτσι μπορώ για παράδειγμα να ελέγξω εάν το 6 είναι μεγαλύτερο από το 5:
(defrule pointless


(test (> 6 5))

=>


(printout t "Six is indeed greater than five" crlf)

)
ΑΣΚΗΣΗ
1. Μεταφράστε τους παρακάτω κανόνες σε CLIPS. 


Αν κάποιος δεν έχει μούσι είναι πωλητής κινητών τηλεφώνων. 


Αν κάποιος έχει μούσι μέχρι 12 ίντσες, είναι φιλόσοφος. 


Αν κάποιος έχει μούσι μεγαλύτερο από 12 ίντσες, είναι θεός. 

2. Με deffacts εισάγετε τα παρακάτω γεγονότα: 


Ο Σωκράτης έχει μούσι 8 ίντσες. 


Ο Del-Boy δεν έχει μούσι. 


Ο Δίας έχει μούσι 18 ίντσες. 
3. Ελέγξτε τους κανόνες σας να βεβαιωθείτε ότι το σύστημά σας βγάζει τα σωστά συμπεράσματα. 

